

MANUAL NEOBOOK

- **AlertBox (VentanaAlerta)**

Sintaxis: AlertBox "título" "mensaje"

Activa una ventana de diálogo conteniendo un título, un mensaje y un botón aceptar.

- **Balloon (Globo)**

Sintaxis: Balloon "mensaje"

Ofrece un globo como el de un tebeo que contiene un mensaje breve.

- **CloseWindow (CerrarVentana)**

Sintaxis: CloseWindow "Nombre de la ventana"

Cierra una ventana abierta previamente con la acción Ventana de imagen o Ventana de texto. Deje el nombre de la ventana en blanco para cerrar todas las ventanas abiertas.

- **Delay (Demora)**

Sintaxis: Delay "milisegundos"

Pausa la ejecución durante un espacio de tiempo determinado en milisegundos (milésimas de segundo).

- **DOSCommand (MandatoDOS)**

-

Sintaxis: DOSCommand "mandato" "parámetros"

Ejecuta una aplicación o un mandato externo DOS o Windows. Los parámetros son objetos que se pasan al programa al lanzar este (por ejemplo, el nombre de un archivo para cargar un documento, opciones para las líneas de mandato, etc.).

- **Exit (Salir)**

Sintaxis: Exit "título" "mensaje"

Le permite salir de la publicación mediante una ventana de diálogo opcional para confirmarlo. Asegúrese de que los campos del título y mensaje permanecen en blanco para que el usuario pueda salir inmediatamente sin visualizar una ventana de diálogo.

- **FileCopy (CopiarArchivo)**

Sintaxis: FileCopy "Nombre del archivo fuente" "Nombre del archivo de destino"

Copia el contenido de un archivo externo a un archivo nuevo.

- **FileDelLine (SuprLíneaArchivo)**

Sintaxis: FileDelLine "nombre del archivo" "número de línea"

Suprime una línea específica de un archivo de datos externo. En el ejemplo que sigue a continuación, se solicita a los lectores si desean suprimir un objeto almacenado en una línea del archivo PARTS.DAT. Su respuesta quedará guardada en una variable llamada "[Res]". De responder Sí (1), la línea quedará suprimida del archivo.

E.g.: MessageBox "Delete" "Delete item #[Objeto]?" "Yes|No" "[Res]"
 If "[Res]" "=" "1"
 FileDelLine "parts.dat" "[Objeto]"
 EndIf

- **FileErase (SuprimirArchivo)**

Sintaxis: FileErase "nombre del archivo"

Suprime un archivo externo.

- **FileInsLine (ArchivoInsLínea)**

Sintaxis: FileInsLine "nombre del archivo" "número de línea a insertar"

Inserta una línea en blanco al archivo de datos externo. El ejemplo que sigue a continuación inserta una nueva primera línea (línea 1) en el archivo PARTES.DAT.

E.g.: FileInsLine "parts.dat" "1"

- **FileLen (LongArchivo)**

Sintaxis: FileLen "nombre del archivo" "nombre de la variable"

Calcula el número de líneas contenidas en un archivo externo. Los datos de la línea quedarán almacenados en la variable especificada por el usuario. El ejemplo que sigue a continuación calcula el número de líneas en el archivo PARTES.DAT y almacena el número de líneas en la variable [Líneas].

E.g.: FileLen "parts.dat" "[Líneas]"

- **FileRead (LeerArchivo)**

Sintaxis: FileRead "nombre del archivo" "número de línea" "nombre de la variable"

Lee los datos del número de línea especificado en un archivo externo. Los datos de la línea quedarán almacenados en la variable especificada. El ejemplo que viene a continuación usa LeerArchivo para construir un menú a partir del contenido del archivo de datos PARTES.DAT. Cada línea se leerá y será colocada en el menú (la variable [Selección]). La función Matemáticas se utiliza para aumentar el número de la línea (la variable [Contar]) que leerá el siguiente objeto del menú, y las opciones del menú se pueden visualizar mediante el mandato MenuEx.

E.g.: SetVar "[ObjMenú]" "x"
 SetVar "[Selección]"
 SetVar "[Contar]" "1"

 While "[ObjMenú]" ">" ""
 FileRead "parts.dat" "[Contar]" "[ObjMenú]"
 If "[ObjMenú]" ">" ""
 If "[Selección]" ">" ""
 SetVar "[Selección]" "[Selección] |"
 EndIf
 SetVar "[Selección]" "[Selección] [ObjMenú]"
 Endif
 Math "[Contar]+1" "0" "[Contar]"
 Endwhile

 MenuEx "-1" "-1" "[Selección]" "[Result]"

- **FileWrite (EscribirArchivo)**

Sintaxis: FileWrite "nombre del archivo" "número de línea" "datos"

Escribe datos al número de línea especificado en un archivo externo. Use "Agregar " en lugar del número de línea para añadir los datos al final del archivo.

- **Find (Buscar)**

Sintaxis: Find "nombre de la primera página|nombre de la segunda página"

Activa la ventana de diálogo buscar texto. Las búsquedas pueden quedar limitadas a un grupo de páginas, a la página actual, o a toda la publicación. Separe los nombres de las páginas con el carácter "|". La búsqueda de las páginas se realizará según la orden solicitada. Utilice "TODAS" para buscar a través de todas las páginas, o bien "ACTUAL" para realizar una búsqueda en la página actual.

- **GotoPage (IrAPágina)**

Sintaxis: GotoPage "nombre de la página"

Pasa directamente a la página especificada de su publicación.

- **GotoFirstPage (IrAPrimeraPágina)**

Sintaxis: GotoFirstPage

Pasa directamente a la primera página de la publicación.

- **GotoLastPage (IrAÚltimaPágina)**

Sintaxis: GotoLastPage

Pasa directamente a la última página de la publicación.

- **GotoNextPage (IrASiguientePágina)**

Sintaxis: GotoNextPage

Pasa a la siguiente página de la publicación.

- **GotoPreviousPage (IrAPáginaAnterior)**

Sintaxis: GotoPreviousPage

Pasa directamente a la página anterior de la publicación.

- **GotoLine (IrALínea)**

Sintaxis: GotoLine "número de línea en el script"

Pasa directamente al número de línea especificada o a la etiqueta en este script de acción y ejecuta el mandato desde ese punto.

- **If (Si)**

Sintaxis: If "primera afirmación" "operador" "segunda afirmación"

Cambia el flujo de la ejecución de la acción según los resultados de una comparación entre dos variables. Si la afirmación es verdadera, la acción ejecuta el script de acción que seguirá en pie hasta que se detecte una acción "FinSi" o "SiNo". De lo contrario, ejecutará el script de acción comenzando con la primera línea que aparece tras el siguiente mandato "SiNo" (opcional) o "FinSi". Los comodines que siguen a continuación se aceptan en los mandatos de la acción Si: =, <, >, <>, <=, >=. En el ejemplo que sigue a continuación se ha configurado un botón para verificar que se haya insertado un nombre en el campo de texto con la variable [Nombre] antes de que se pueda acceder a la siguiente página. De no haberse insertado ningún nombre, el ordenador emitirá un aviso sonoro y se activará un mensaje de Alerta.

E.g.: If "[Nombre]" "<>" ""

```
GotoNextPage
    Else
        SoundBuzzer
        MsgBox "¡Inserte el nombre!" "Inserte su nombre para continuar."
    EndIf
```

También puede usar esta función para detectar la correcta inserción de una contraseña.

E.g.: If "[Contraseña]" "=" "ST0172"
GotoNextPage
Else
SoundBuzzer
MsgBox "Contraseña Inválida" "La contraseña no es correcta."
EndIf

- **ImageWindow (VentanaImagen)**

Sintaxis: ImageWindow "título" "pixels desde la izquierda" "pixels desde arriba" "nombre del archivo"

Muestra un archivo de imagen bitmap en una ventana desplazable. Use 1 en la parte izquierda y superior para centrar la ventana en pantalla. Tenga en cuenta que al usar la herramienta Insertar acción, puede colocar la ventana pulsando el botón Colocar/Escala ubicado a la derecha del diálogo.

- **InternetLink (EnlaceInternet)**

Sintaxis: InternetLink "Dirección URL"

Inicia el buscador de Internet y accede directamente a la dirección URL de una página o FTP especificada. (Para que esta acción se pueda ejecutar, el usuario final deberá tener instalado en su ordenador el software del buscador Internet y disponer de una cuenta/conexión a Internet).

- **Math (Matemáticas)**

Sintaxis: Math "cálculo" "puntos decimales" "variable de resultados"

Realiza un cálculo matemático y almacena los resultados en una variable. Puede usar la mayoría de las funciones estándar tales como +, -, *, /, sen, cos, tan, raíz cuadrada, raíz cúbica, redondear y trunc. También puede crear complejas ecuaciones usando paréntesis y variables definidas por el autor. Tenga en cuenta que las variables siempre aparecen contenidas dentro de paréntesis cuadrados "[]" y se pueden utilizar en cualquier lugar de NeoBook conjuntamente con texto. En sus cálculos matemáticos podrá utilizar las operaciones que aparecen a continuación: +, -, *, /, abs, sin, cos, tan, raíz cuadrada, raíz cúbica, redondear, trunc, (,). En el ejemplo que aparece a continuación, la función Matemáticas usa una variable [AlquilerMensual] para completar la computación. [AlquilerMensual] se puede configurar usando la opción FijarVar o asociarlo con un botón Inserción texto para permitir que el usuario pueda insertar un número. El resultado contiene dos decimales (el segundo parámetro) y está guardado en la variable [AlquilerSemanal]. La variable [AlquilerSemanal] se puede visualizar en un Título de texto o en una Ventana de alerta (por ejemplo, puede usar el mandato: MsgBox "Resultado" "La respuesta es [AlquilerSemanal]").

E.g.: Math "([AlquilerMensual]*12)/52" "2" "[AlquilerSemanal]"

- **MCICommand (MandatoMCI)**

Sintaxis: MCICommand "mandato"

Envía un mandato al interfaz control de medios Windows (MCI). Ofrece un acceso a alto nivel a varios dispositivos multimedia. Si desea obtener más información en cuanto a los mandatos MCI, véase la documentación SDK de Microsoft Windows. El mandato MCI dará un cero o un número erróneo como resultado en la variable global [ResultadoMCI].

- **Menu (Menú)**

Sintaxis: Menu "pixels desde la izquierda" "pixels desde arriba" "primera|segunda selección"

Activa un menú en pantalla. Se ejecutará el mandato de acción en una de las líneas que sigue a continuación y que corresponda a la opción seleccionada. Por ejemplo, si su menú dispone de tres selecciones, la primera selección seleccionará la línea que aparece inmediatamente después de la línea de acción del menú; la segunda usará la segunda acción después de la línea del menú; la tercera usará la acción en la tercera línea después del menú.

Use 1 en la parte izquierda y superior para centrar la ventana en pantalla. Separe las selecciones con el carácter "|". Tenga en cuenta que al usar la herramienta Insertar acción, puede colocar el menú pulsando el botón Colocar ubicado en la parte derecha del diálogo. El ejemplo que sigue a continuación le ofrece al lector tres opciones y a continuación activa una Ventana de alerta que corresponde al objeto seleccionado:

E.g.: Menu "-1" "-1" "Manzanas|Naranjas|Uvas"
AlertBox "" "Ha seleccionado Manzanas"
AlertBox "" "Ha seleccionado Naranjas"
AlertBox "" "Ha seleccionado Uvas"

- **MenuEx (MenúEx)**

Sintaxis: MenuEx "pixels desde la izquierda" "pixels desde arriba" "selección 1|selección 2" "nombre de la variable"

Muestra un menú de opciones a medida y guarda la opción seleccionada como un número (1 equivale a primero, 2 a segundo... o cero a ninguno) en una variable. Debido a que la respuesta se guarda como una variable y ésta se puede escribir a un archivo o bien se puede utilizar en cálculos, conviene usar este estilo de menú para preguntas, formularios, etc. Use 1 para centrar el menú en la parte superior izquierda de la pantalla. Separe las opciones con el carácter "|". Recuerde que al usar la herramienta Insertar acción, puede colocar el menú pulsando el botón Colocar en la parte derecha del diálogo. En el ejemplo que sigue a continuación se le ofrece al lector tres opciones. Al realizar una selección, el número del objeto seleccionado se colocará en la variable [Resultado] y se activará la Ventana de Alerta correspondiente:

E.g.: MenuEx "-1" "-1" "Manzanas|Naranjas|Uvas" "[Resultado]"

```
If "[Resultado]" "=" "1"  
AlertBox "" "Ha seleccionado manzanas"  
EndIf  
  
If "[Resultado]" "=" "2"  
AlertBox "" "Ha seleccionado naranjas"  
EndIf  
  
If "[Resultado]" "=" "3"  
AlertBox "" "Ha seleccionado uvas"  
EndIf
```

- **MessageBox (VentanaDeTexto)**

Sintaxis: MessageBox "título" "mensaje" "selección 1|selección 2" "nombre de la variable"

Ofrece un mensaje personalizado y guarda la opción seleccionada como número (1 para el primero, 2 para el segundo... cero para ninguno) en una variable. Separe las selecciones con el carácter "|". En el ejemplo que sigue a continuación se le solicita al lector que determine su humor (guardado en la variable [Humor]), y se le ofrece la Ventana de Alerta correspondiente:

E.g.: MessageBox "Buenos días" "¿Cómo está hoy?" "Bien|Así así|Mal" "[Humor]"

```
If "[Humor]" "=" "1"
    MsgBox "" "¡Genial!"
EndIf

If "[Humor]" "=" "2"
    MsgBox "" "Espero que mejore su humor."
EndIf

If "[Humor]" "=" "3"
    MsgBox "" "Lo siento."
EndIf
```

- **PlayCartoonFile (ReproducirArchivoAnimado)**

Sintaxis: PlayCartoonFile "empezar desde la izquierda" "empezar desde arriba" "finalizar a la izquierda" "finalizar arriba" "nombre del archivo" "tiempos" "demora" "modalidad"

Reproduce un archivo de animación generado por NeoBook en el módulo NeoToon. Especifique las coordenadas de principio y fin de la pantalla para crear una vía por la que se desplace la animación (use 1 para centrar la animación en la página), seguido por un nombre de archivo y el número de veces que la animación debe entrar en bucle desde el punto inicio y punto final. Use el control Velocidad en el diálogo Insertar mandato de acción para indicar la demora (en milisegundos) que desea colocar entre cada diapositiva de la animación durante su reproducción. Use el parámetro final para indicar si ha de seguir ejecutándose el script (modalidad "Normal") al inicio de la animación, si se ha de reproducir la animación de forma continua entre el punto inicio y el punto final usando la modalidad "Bucle" o si ha de completarse la animación (modalidad "Espera") antes de que se realice la siguiente acción. Puede usar DetenerMedio en otro botón para detener la animación actual.

Nota: Al usar la herramienta Insertar mandato de acción, puede colocar una trayectoria que deberá seguir la animación pulsando el botón Colocar/Escala a la derecha del diálogo.

- **PlayFlicFile (ReproducirArchivoFlic)**

Sintaxis: PlayFlicFile "desde la izquierdat" "desde arriba" "ancho" "alto" "nombre del archivo" "repeticiones" "modalidad"

Ejecuta un archivo de animación en formato Flic (FLI/FLC). Si desea centrar la página, utilice -1 como valor superior e izquierdo. Si desea ejecutarlo al tamaño original, utilice -1 para el ancho y el alto. En el campo de las veces que se ejecutará la animación, especifique el número de veces que esto ocurrirá. Utilice el parámetro final para indicar si el script continuará ejecutándose (modalidad "Normal") una vez que la animación comience, si la animación se ejecutará de forma continua desde el inicio hasta el fin, (modalidad "Bucle") o si la animación debe completarse en su totalidad antes de ejecutarse la próxima acción (Modalidad "Espera"). Utilice DetenerMedio si desea detener la animación que se esté ejecutando en la actualidad.

Recuerde que al utilizar la herramienta Insertar mandato de acción, debe indicar la escala de animación pulsando el botón Ubicación/escala ubicado a la derecha del diálogo.

- **PlaySoundFile (ReproducirArchivodeSonido)**

Sintaxis: PlaySoundFile "nombre del archivo" "tiempo"

Reproduce un archivo de sonido (WAV) o de música (MDI). Use la modalidad "Normal" para reproducir el archivo de sonido y pasar a la siguiente selección del script. Use la modalidad " Espera" para dar el archivo de sonido por finalizado antes de continuar. Use la modalidad " Bucle" para reproducir un bucle continuo. Recuerde que sólo se puede reproducir un archivo del mismo tipo de sonido simultáneamente. Use DetenerMedio para detener el archivo de sonido.

- **PlayTone (ReproducirTono)**

Sintaxis: PlayTone "frecuencia"

Reproduce un tono (especificado como un número en Herzios) en el altavoz interno del ordenador.

- **PlayVideoFile (ReproducirArchivoVideo)**

Sintaxis: PlayVideoFile "pixels desde la izquierda" "pixels desde arriba" "ancho" "alto" "nombre del archivo" "modalidad"

Reproduce un archivo de vídeo o animación (AVI/MOV/MPG). Use 1 en los valores izquierda y superior para centrar la página. Use el parámetro final para indicar si ha de seguir ejecutándose el script (modalidad "Normal") una vez iniciada la animación, si el vídeo ha de repetirse de forma continua (modalidad "Bucle") o si la animación debe completarse (modalidad "Espera") antes de que se ejecute la siguiente acción. Use DetenerMedio para detener la reproducción de vídeo.

Recuerde que al usar la herramienta Insertar acción, puede colocar/ajustar el vídeo pulsando el botón Colocar/Escala a la derecha del diálogo.

- **PopupImage (Imagen)**

Sintaxis: PopupImage "pixels desde la izquierda" "pixels desde arriba" "nombre del archivo" "demora" "efecto del nombre" "velocidad"

Muestra un archivo de imagen bitmap con un efecto especial opcional y espera hasta que se pulse el botón del ratón o una tecla o bien demora el número de milisegundos. Use "-1" en los valores izquierda y superior para centrar la imagen en la pantalla. Use "0" (cero) en el campo demora para esperar hasta que el usuario pulse el botón del ratón o una tecla. Use los valores de 0 (rápido) a 10 (lento) para ajustar la velocidad del efecto.

Los efectos especiales incluyen: "Ninguno", "Disolver", "Desplazamiento a la izquierda", "Desplazamiento a la derecha", "Desplazamiento hacia arriba", "Desplazamiento hacia abajo", "Explosión", "Implosión", "Entrelazado horizontal", "Entrelazado vertical", "División horizontal", "División vertical", "Limpiar izquierda", "Limpiar derecha", "Limpiar arriba" y "Limpiar abajo".

Recuerde que al usar la herramienta Insertar acción, puede colocar y ajustar la imagen pulsando el botón Colocar/Escala a la derecha del diálogo.

- **PrintDataFile (ImprimirArchivoDatos)**

Sintaxis: PrintDataFile "texto del encabezado" "texto pie de página" "nombre del archivo"

Imprime un archivo de datos externo ASCII (normalmente creado con la acción EscribirArchivo).

- **PrintImageFile (ImprimirArchivoImagen)**

Sintaxis: PrintImageFile "nombre del archivo" "factor de escala"

Imprime un archivo de imagen bitmap.

- **PrintPage (ImprimirPágina)**

Sintaxis: PrintPage "nombre de la página"

Imprime la página especificada de esta publicación.

- **PrintTextFile (ImprimirArchivoTexto)**

Sintaxis: PrintTextFile "texto del encabezado" "texto pie de página" "nombre del archivo"

Imprime un archivo de texto ASCII.

- **Random (Aleatorio)**

Sintaxis: Random "valor máximo" "nombre de la variable"

Esta acción genera un número aleatorio entre cero y el máximo especificado. El número aleatorio resultante se puede guardar en una variable. En el ejemplo que sigue a continuación, el ancho y el alto de la publicación han sido determinados, y la acción Aleatorio se utiliza para colocar una imagen en una ubicación a dispar (generando variables para las coordenadas "X" "Y" horizontal/vertical y la demora de "L"ongitud) de la pantalla:

E.g.: :Inicio

```
SystemInfo "PubWidth" "[PW]"
SystemInfo "PubHeight" "[PH]"
```

```
Random "[PW]" "[X]"
Random "[PH]" "[Y]"
Random "5000" "[L]"
```

```
PopupImage "[X]" "[Y]" "C:\IMAGES\FLOWER.BMP" "[L]" "Dissolve" "0"
```

```
GotoLine ":Inicio"
```

- **ReturnToPage (VolverAPágina)**

Sintaxis: ReturnToPage

Vuelve a la última página visitada por el usuario.

- **SearchStr (BuscarStr)**

Sintaxis: SearchStr "buscar" "número de línea" "nombre de la variable"

Busca caracteres dentro de un texto y guarda la posición del primer carácter (o cero si no consigue ubicarlo) en una variable. El ejemplo que sigue a continuación busca los espacios en blanco dentro de la variable [Nombre] y guarda la ubicación del espacio en blanco en la variable [PosEspacio], y a continuación usa la acción SupStr para borrar el espacio en blanco almacenando los resultados modificados en la variable [Nombre]. Seguirá realizando esta acción hasta que desaparezcan todos los espacios dentro de [Nombre]:

```
E.g.: SearchStr " " "[Nombre]" "[PosEspacio]"
      While "[PosEspacio]" ">" "0"
          SearchStr " " "[Nombre]" "[PosEspacio]"
          StrDel "[Nombre]" "[PosEspacio]" "1" "[Nombre]"
      EndWhile
```

- **SetFont (FijarFuente)**

Sintaxis: SetFont "nombre de la fuente" "tamaño" "estilo"

Fija la fuente utilizada para imprimir y visualizar los archivos de texto. Use este mandato con las acciones ImprimirArchivoTexto y VentanaTexto.

- **SetVar (FijarVar)**

Sintaxis: SetVar "nombre" "valor"

Si desea resetear el valor de la variable a cero, use "0" (carácter cero). Para resetear una variable de texto en blanco, utilice "" (sin teclear nada entre las comillas). En el ejemplo que sigue a continuación, la variable [Precio] está fijada a "49.95" y la variable [Nombre] ha quedado fijada a "Ralph Jones":

E.g.: SetVar "[Precio]" "49.95"
SetVar "[Nombre]" "Ralph Jones"

También puede usar FijartVar para modificar o combinar las variables ya existentes:

E.g.: SetVar "[NombreCompleto]" "[Apellido], [NombredePila] [PrimerApellido]"
SetVar "[VisualizarPrecio]" "Ptas.[Precio]"

- **SoundBuzzer (AvisoSonoro)**

Sintaxis: SoundBuzzer

Reproduce un tono de error a través del altavoz interno del ordenador.

- **StickyNote (Nota)**

Sintaxis: StickyNote "pixels desde la izquierda" "pixels desde arriba" "mensaje" "demora"

Muestra una nota conteniendo un mensaje en pantalla. Use 1 en los valores izquierdo y superior para centrar la nota en la página. Use "0" (cero) en el campo demora para esperar hasta que el usuario pulse una tecla.

Nota: Al usar la herramienta Insertar acción, puede colocar la nota pulsando el botón Colocar a mano derecha del diálogo.

- **StopMedia (DetenerMedio)**

Sintaxis: StopMedia "nombre del archivo"

Detiene la reproducción de archivos multimedia (sonido, vídeo o animaciones) de forma individual o total. Deje el nombre del archivo en blanco para detener todos los archivos multimedia. Utilice "Sonido" en el campo del nombre del archivo para detener toda la reproducción audio. Use "Vídeo" para detener la reproducción vídeo.

- **StrDel (SupStr)**

Sintaxis: StrDel "string fuente" "posición de inicio" "número de caracteres que desea suprimir" "nombre de la variable"

Suprime los caracteres de una línea, y almacena el resultado como una variable. En el ejemplo que sigue a continuación, la variable [Palabra] contiene 23 letras y debemos abreviarlo a un espacio de 20 letras. Esta acción se coloca en la posición 21 y suprime los tres caracteres guardando los resultados en la variable [MostrarPalabra]:

E.g.: StrDel "[Palabra]" "21" "3" "[MostrarPalabra]"

- **StrIns (InsStr)**

Sintaxis: StrIns "string fuente" "string destino" "posición donde insertar" "nombre de la variable"

Inserta una línea dentro de otra en la posición especificada y almacena la serie editada dentro de una variable. En el ejemplo que sigue a continuación, supongamos que la variable [Día] contiene un valor de 22, y la variable [MesAño] contiene "Octubre 1998". El resultado sería la variable [DíaMesAño] conteniendo la variable "22, octubre 1998":

E.g.: StrLen "[Día], " "[MesAño]" "1" "[DíaMesAño]"

- **StrLen (StrLen)**

Sintaxis: `AlertBox "nombre de la serie" "nombre de la variable"`

Calcula la longitud de una línea de texto y almacena los resultados dentro de una variable. En el ejemplo que sigue a continuación, la variable [Nombre] contiene la serie "Haroldo". Debido a que esta serie contiene siete caracteres, la acción del ejemplo colocará un valor de "7" dentro de la variable [Tamaño]:
E.g.: `StrLen "[Nombre]" "[Tamaño]"`

- **SubStr (SubStr)**

Sintaxis: `SubStr "nombre de la serie" "carácter de inicio" "número de copias" "nombre de la variable"`

Copia una porción de la línea de texto y coloca los resultados en una variable. En el ejemplo que sigue a continuación, la variable [ST] dispone de un valor de "ABCDEFGH", por lo que esta acción entregará el valor "CD" (es decir, la C es el tercer carácter y la acción pide dos caracteres desde C):
E.g.: `SubStr "[St]" "3" "2" "[NuevoSt]"`

- **SystemInfo (InfoSistema)**

Sintaxis: `SystemInfo "tipo de información" "nombre de la variable"`

Obtiene información acerca de este ordenador y guarda los resultados en una variable. El tipo de información incluye: PubAuthor, PubColors, PubDir, Pub Height, PubWidth, PubTitle, ScreenColors, ScreenHeight, Screen Width, SystemDir, TempDir, WindowsDir, UserName. En el ejemplo que sigue a continuación, la resolución de color del ordenador del usuario se compara a la resolución de color utilizada en la publicación. Aparecerá un mensaje si el ordenador del usuario dispone de una resolución inferior:

E.g.: `SystemInfo "ScreenColors" "[ModalidadPantalla]"
SystemInfo "PubColors" "[ModalidadPub]"

If "[PubMode]" ">" "[ModalidadPantalla]"
AlertBox "¡Aviso!" "Este programa requiere [ModalidadPub] colores para una
correcta visualización. Windows está funcionando en modalidad [ModalidadPantalla] colores| por lo que
no verá este programa bajo condiciones |óptimas."
EndIf`

- **TextWindow (VentanaTexto)**

Sintaxis: `TextWindow "título de la ventana" "pixels desde la izquierda" "pixels desde arriba" "nombre del archivo"`

Coloca un archivo de texto en una ventana movable. Use 1 en los valores izquierdo y superior para centrar la página o especifique las coordenadas de la página. Recuerde que al usar la herramienta Insertar acción, puede colocar la ventana pulsando el botón Colocar a la derecha del diálogo.

- **While (Mientras)**

Sintaxis: `While "afirmación variable" "operador" "valor "`

Repite una serie de acciones hasta obtener una condición especificada (valor). Los operadores condicionales que siguen a continuación se aceptarán en los mandatos de acción Mientras: =, <, >, <>, <=, >=. En el ejemplo que sigue a continuación se reproducen 100 tonos usando la acción ReproducirTono. El sonido comienza a una frecuencia de 0 Hertzios y se va incrementando a 25 Hertzios hasta reproducir los 100 tonos en secuencia:

E.g.: `SetVar "[Contar]" "0"
SetVar "[Frecuencia]" "0"

While "[Contar]" "<" "100"
Math "[Contar]+1" "0" "[Contar]"
Math "[Frecuencia]+25" "0" "[Frecuencia]"
PlayTone "[Frecuencia]"
EndWhile`